Easy Language

Traders trade with TradeStation®

PaintBar Studies

Ex-20 MomentumPositive

Learning objective: Writing a PaintBar study using the PlotPB statement

Write a PaintBar that marks those bars on which the Momentum is greater than 0.

Ex-20 MomentumPositive

[Update Value Intra-Bar (Tick by Tick)]

```
Input: Price(Close), Length(10);

Vars: Mom(0);

Mom = Momentum(Price, Length);

If Mom > 0 then begin
      PlotPB(High, Low, "MomPos");
      Alert;
end
else
      NoPlot(1);
//NoPlot unpaints the bar if the condition changes during the bar
```

Creating Trading Strategies

- אסטרטגית מסחר הינה מערכת של כללי מסחר המבצעת כניסה ויציאה מתוכנתים לפוזיציות מסחר.
- TradeStation מספקת כלים ומנגנון להרצת BackTesting היסטורי של האסטרטגיה, המאפשרת לבדוק ולהעריך את ביצועים היסטוריים, לבצע אופטימיזציה, ואוטומציה בביצוע פקודות בשלב מסחר אמיתי עם האסטרטגיה
- אסטרטגיה מכילה פעולת מסחר אחת לפחות. גרף - Chart יכול להכיל ולבצע מספר אסטרטגיות בו זמנית.

Creating Trading Strategies

- Buy [("Order Name")] [Number of Shares/Contracts] [Order Action];
- SellShort [("Order Name")] [Number of Shares] Execution Method ;
- דוגמא:

if Average(Close,9) Crosses above the Average(Close,18) then
Buy ("MA2CrossLE") next bar 100 shares at Market;

if Average(Close,9) Crosses Below the Average(Close,18) then
Sell Short("MA2CrossSE") next bar 100 shares at Market;

Creating Trading Strategies



Creating Trading Strategies

Strategy Engine Calculation

- **On Bar Close** -האסטרטגיה מחושבת רק פעם אחת עבור כל בר, בסגירה של הבר. במידה ומתקיימים התנאים פקודות מסחר מיוצרות עבור הבר הבא, ומבוטלות במידה ולא הופעלו במשך הבר הבא.
- **Intra Bar** - מבחינה הסטורית ,האסטרטגיה מחושבת פי הרזולוציה שנקבעה בפרמטר Look Bar Back-Testing Inside
- בהרצת האסטרטגיה בזמן האמת, האסטרטגיה מחושבת על בסיס Tick . במידה ומתקיימים התנאים פקודות מסחר מיוצרות עבור ה Tick הבא, ומבוטלות במידה ולא הופעלו במשך Tick הבא.

Strategy Properties



Strategy Properties

Costs/Capitalization

- Commissions & Slippage
- Initial Capital
- Interest Rate

Back-testing Resolution

- **Look-Inside-Bar Back-Testing**
- **Max Number of bars study will reference**

Strategy Properties

Position Limits

- Num of Entry orders in the same direction
- Position Size – Max Shares/Contracts/Units

Trade Size

- Fixed Shares/Contracts/Units vs $ Per Trade

Strategy Automation



Strategy Order Syntax

- **Buy:** Establish or add to a long position. Any existing short position will be covered entirely before the long position is established. Two orders are generated.
- **SellShort** or **Sell Short**: Establish or add to a short position. Any existing long position will be liquidated entirely before the short position is established. Two orders are generated.

Example: Our Trend Following Strategy

Strategy Order Syntax

- **Sell**: Liquidates a long position only. Can never establishes a short position.
- **BuyToCover or Buy To Cover**: Cover a short position only.

Strategy Order Syntax

Order Actions

- **next bar at Market**: Market order at the open of the next bar or the next tick.
- **next bar Stop**: Market order on the next bar when the stop price is reached. All stop orders in Easy-Language are Stop Market.
- **next bar Limit:** Limit order on the next bar if the limit price is reached.

Strategy Order Syntax

Order Actions

- **this bar on Close**: Market order on the close of this bar, generally used for historical backtesting purposes only. This will not generate a market on close order.

Usage Examples:

- Buy next bar at Market; Buy ("**Le1**") next bar at Market;
- Sell Short next bar 50 Limit; Sell Short ("**Sh1**") next bar 50 Limit;
- Sell next bar 50 Stop;
- Buy to Cover this bar on Close;

Orders – Signal Names

Usage Example:

Buy ("L1LE") this bar on Close;
SellShort("L1SE") next bar at Market;



Ex-23 Breakout

Learning objective: Writing a Strategy; stop order syntax.

Description: Write a Strategy that buys (long) on a breakout above the highest high of the preceding 8 bars; it sells short on a break down below the lowest low of the preceding 8 bars.

Ex-23 Breakout

TradeStation Development Environment - *23 Breakout : Strategy*

File Edit View Build Debug Tools Window Help

23 Breakout : Strategy

```
{
   Strategy:      *23 BreakOut

   This Strategy buys (long) on a breakout above the highest
   high of the preceding 8 bars;
   It sells short on a break down below the lowest low of
   the preceding 8 bars.
}

Input: Length(8);

Vars: BuyPx(0), SellPx(0);

 BuyPx = Highest(High, Length) + .02;
 SellPx = Lowest(Low, Length) - .02;

 //Stop order to buy above "BuyPx";  Signal name is in parentheses following verb
 Buy ("Brk LE") next bar at BuyPx Stop;
 //Stop order to SellShort below "SellPx";  Signal name is in parentheses following verb
 SellShort ("Brk SE") next bar at SellPx Stop;
```

Ex-23 Breakout



Strategy Info – Market Position

MarketPosition(*N*) returns whether the strategy is currently flat, short, or long on the current bar or for *N* closed positions ago. MarketPosition return values are:

- -1 for a short position.
- 1 for a long position.
- 0 for flat (no position).

Usage Example:

if **MarketPosition** = 0 then
    Buy next bar at Market;

Strategy Info – Market Position

Historical Reference of Strategy Position Reserved Words

Usage Example:

Vars: MP(0);
MP = MarketPosition;
if **MP[2]** = -1 AND **MP[1]** = 0 then
    Buy next bar at Market;

Strategy Info – Entry Price

EntryPrice – returns the entry price for the current position.

It can also report what the entry price was *N* closed positions ago – **EntryPrice(N)**.

Usage Example:

```
if MarketPosition = 1 then
        Sell next bar at EntryPrice - .10 Stop;
```

Strategy Info – Bars Since Entry

BarsSinceEntry – returns the number of bars from the entry bar for the current position.

It can also report how many bars from the entry bar of *N* closed positions ago – **BarsSinceEntry(N)**

Strategy Info – Bars Since Entry

BarsSinceEntry returns the number of bars from the entry bar for the current position.

Usage Example 1:

```
if MarketPosition = 1 AND BarsSinceEntry > 5 then
        Sell next bar at Market;
```

Usage Example 2:

```
if MarketPosition = 1 AND BarsSinceEntry > 5 then
        Sell next bar at Low[BarsSinceEntry] Stop;
```

Additional Strategy Position Info

- AvgEntryPrice
- BarsSinceExit
- Current Shares
- Current Contracts
- EntryDate
- EntryTime
- ExitDate
- ExitTime

Ex-24 Mov Avg Cross

Learning objective: Writing a Strategy; using Strategy Position Reserved Words.

Description: Write a Strategy that takes long and short positions based on the crossovers of two moving averages.
A position is closed when it has been held at least a minimum number of bars and is not making at least a minimum profit.

Ex-24 Mov Avg Cross

24 Mov Avg Cross : Strategy

```
{
    Strategy:      *24 Mov Avg Cross

    This Strategy takes long and short positions based on the crossovers of two moving averag
    A position is closed when it has been held at least a minimum  number of bars and is not
    making at least a minimum profit.
}

Input: ShortLen(9), LongLen(18), MinHold(8), MinProf(50);

Vars: ShortMA(0), LongMA(0);

 ShortMA = Average(Close, ShortLen);
 LongMA = Average(Close, LongLen);

 If ShortMA crosses over LongMA then
     Buy next bar at market;
 If ShortMA crosses under LongMA then
     SellShort next bar at market;

If BarsSinceEntry > MinHold and OpenPositionProfit < MinProf then begin
     Sell ("EL-M1") next bar at market;
     //Sell and BuyToCover are in a Block If/then since they are dependent
     BuyToCover ("ES-M2") next bar at market;
     //on the same conditions, and mutually exclusive
end;
```

Ex-24 Mov Avg Cross

Strategy Performance Info – NetProfit

NetProfit returns the cumulative net profit or loss for all closed trades in the chart; this is the closed trade equity curve value for each bar. The value will either be positive, negative, or zero.

Usage Example :

```
Vars: TradeSize(0);

TradeSize = 1000;
TradeSize = TradeSize +( NetProfit / Close);
Buy next bar TradeSize Shares at Market;
```

Strategy Performance Info – OpenPositionProfit

OpenPositionProfit returns net profit or loss for the current open position.

Adding **OpenPositionProfit** to **NetProfit**, derives the same value as the detailed equity curve on a bar-by-bar basis.

The value will either be positive, negative, or zero.

Usage Example:

```
Input: ProfitExit(200);

if OpenPositionProfit >= ProfitExit then begin
        Sell next bar at Market;
        BuyToCover next bar at Market;
end;
```

Additional Strategy Performance Info

- GrossProfit
- GrossLoss
- NumWinTrades
- NumLosTrades
- PercentProfit
- Total Trades

Ex-25 Momentum Cross

Learning objective: Using user-declared true/false variables; MRO-Most Recent Occurrence Function

Description: Write a Strategy that:

- Buys (long) when Momentum crosses over 0, as long as it has not crossed under 0 within the last 4 bars; it sells (liquidates long positions) if Momentum declines on 2 consecutive bars;
- Sells short when Momentum crosses under 0, as long as it has not crossed over 0 within the last 4 bars; it buys to cover (covers short positions) if Momentum rises on 2 consecutive bars.

Ex-25 Momentum Cross

MRO-Most Recent Occurrence Function

MRO (Function)

Disclaimer

The MRO (Most Recent Occurrence) function returns the number of bars ago the specified expression was True. Or, if the specified expression did not occur within the last *x* number of bars, the function informs you of such.

Syntax

```
MRO(Test, Length, Instance)
```

Returns (Integer)

A numeric value containing the number of bars ago that the specified Expression was True; -1 if Expression was not found to be True within the last Length bars.

Parameters

Name	Type	Description
Test	TrueFalse	Sets the true/false expression to check for (that is, Close > Open).
Length	Numeric	Sets the number of bars to check.
Instance	Numeric	Sets which occurrence, for example, 1 = most recent, 2 = 2nd most recent, and so on.

Ex-25 Momentum Cross

25 Momentum Cross : Strategy

```
Input: Length(10);

Vars: Mom(0), BullCx(false), BearCx(false);
//BullCx and BearCx are user-declared true/false Variables

Mom = Momentum(Close, Length);
BullCx = Mom crosses over 0;
//A true false expression must be assigned to a true/false Variable
BearCx = Mom crosses under 0;

If BullCx and MRO(BearCx,4,1) = -1 then
//MRO function checks that there was no BearCx in the last 4 bars
    Buy next bar at Close of this bar limit;

If BearCx and MRO(BullCx,4,1) = -1 then
//MRO function checks that there was no BullCx in the last 4 bars
    SellShort next bar at Close of this bar limit;

If Mom < Mom[1] and Mom[1] < Mom[2] then
//Exit long position after 2 consecutive bars of weaker momentum
    Sell next bar at market;

If Mom > Mom[1] and Mom[1] > Mom[2] then
//Exit short position after 2 consecutive bars of stronger momentum
    BuyToCover next bar at market;
```

Ex-25 Momentum Cross

Built-in Stops

- **SetDollarTrailing** – sets an exit stop a fixed number of dollars away from the peak profit.
- **SetPercentTrailing** – sets an exit stop a fixed percent of the peak profit away from the peak profit, after a minimum profit is achieved.

Built-in Stops

- **SetProfitTarget** – sets an exit order at a fixed dollar profit target.
- **SetStopLoss** – sets a stop loss order at a fixed dollar risk from entry.
- **SetBreakEven** – sets an exit stop at the entry price, after a minimum profit is achieved.

Built-in Stops

Usage Example:

```
Vars: MoValue(0);

MoValue = Momentum(Close, 10);
if MoValue crosses over 0 then
   Buy next bar at Market;
SetStopLoss(100);
SetProfitTarget(100);
```

Built-in Stops

SetStopPosition – exit is calculated for the entire position in dollars.

SetStopShare or **SetStopContract** – exits are calculated per share or contract.

Usage Example:

```
Inputs: StopAmt(1), ProfitAmt(1);

SetStopShare;
SetStopLoss(StopAmt);
SetProfitTarget(ProfitAmt);
```

Trailing Stop הכנסת

Ex-26 Key Reversal

Write a strategy that uses key reversals up and key reversals down to identify entry points.

Declare and assign Variables for key reversals up and down.

Have the strategy enter a limit order to buy on the bar following a key reversal up, at a limit price better than the current bar's close.

Ex-26 Key Reversal

Have the strategy enter a limit order to sell short on the bar following a key reversal down, at a limit price better than the current bar's close.

Declare an Input for the number of points above or below the reversal
bar's close to set the limit order prices; have the Input default to 5 points.

Ex-26 Key Reversal

```
Input: LimitPoints(.05);

Vars: RevUp(false), RevDown(false);

 RevUp = Low < Low[1] and Close > Close[1];  //True/false Variable
 RevDown = High > High[1] and Close < Close[1];

 If RevUp then
    Buy next bar at Close of this bar - LimitPoints limit;  //Posi
                                                 //the bar after the
 If RevDown then
    SellShort next bar at Close of this bar + LimitPoints limit;
```

Commentary (Reserved Word)

This reserved word sends the expression (or list of expressions) to the Analysis Commentary window for whatever bar is selected on the price chart.

Example:

```
Commentary("This is one line of commentary") ;
Commentary("The 10-bar avg = ", Average(Close, 10), Newline);

Commentary("Fast_Avg=", Fast_Avg, "Slow_Avg=", Slow_Avg),
Newline);
```

RSI - מדד העוצמה היחסית

- RSI - Relative Strength Index . פותח בשנת 1978 על ידי ולס ויילדר.
- המדד אומד את העוצמה בין ממוצע הרווח לבין ממוצע ההפסד במחיר נייר הערך על פני מספר תקופות נתון (וילדר המליץ על תקופה של 14)
- מדד ה RSI נע על סקלת הערכים בין 0-[שיא החולשה] ל 100-[שיא העוצמה] כאשר קו 50 הינו קו האמצע.

$$RSI = 100 - \frac{100}{1 + RS}$$

$$RS = \text{Average Gain} / \text{Average Loss}$$

- Average Gain = [(previous Average Gain) x 13 + current Gain] / 14.
- Average Loss = [(previous Average Loss) x 13 + current Loss] / 14.

RSI - מדד העוצמה היחסית - שימושים

- RSI- 70 ומעל מכונה רמת קנית יתר (overbought) ומצביע על התחזקות המניה.
- RSI - 30 ומתחת מכונה רמת מכירת יתר (oversold) ומצביע על החלשות המניה.
- עליה במדד מעל 30 נחשבת לאיתות שורי, וירידה מתחת ל 70 נחשבת לאיתות דובי.
- עליה מעל 50 בערך המדד נחשבת לאישוש לנטיה שורית (עוצמת הרווחים הממוצעת גדולה מעוצמת ההפסדים הממוצעת) וירידה מתחת ל 50 לאישוש לנטיה דובית.

RSI - מדד העוצמה היחסית - שימושים

- התבדרות (divergence) בערכי המדד לבין ערכי נייר הערך באזורי קנית/מכירת יתר מהווה איתות אמין יותר להיפוך מגמה .

DELL - גרף

RSI - מדד העוצמה היחסית

RSI - מדד העוצמה היחסית- AAPL

RSI - מדד העוצמה היחסית-GOOG

RSI - מדד העוצמה היחסית- EMC

RSI - מדד העוצמה היחסית - DNA



Ex-28 RSI Overobught Oversold

- Learning objective: Using Built-in Stops in custom Strategies; using HighestBar and LowestBar Functions.
- Description: This strategy takes long (short) positions when the oversold (overbought) RSI begins to turn. For longs, RSI must be below 50 and a 7-bar low must have been made at least 3 bars ago.
- Shorts are the reverse. Built-in Stops have been added.

Ex-28 RSI Overobught Oversold

```
Input: Length(14), StopAmt(50), BEAmt(50), TrlgAmt(100);

Var: xRSI(0);

xRSI = RSI(Close, Length);

If xRSI < 50 and LowestBar(xRSI, 7) >= 3 then
   Buy next bar at market;

If xRSI > 50 and HighestBar(xRSI, 7) >= 3 then
   SellShort next bar at market;

SetStopLoss(StopAmt);
SetBreakeven(BEAmt);
SetDollarTrailing(TrlgAmt);
```